

Interfata serială standard

Portul serial este de tip complet duplex, ceea ce înseamnă că el poate transmite și recepționa simultan. El este de asemenea receptor bufferat, ceea ce înseamnă faptul că el poate începe recepția celui de-al doilea octet, înainte ca octetul recepționat anterior să fi fost citit din registru (totuși, dacă primul octet nu a fost încă citit până în momentul recepționării complete a celui de-al doilea octet, atunci unul din cei doi octeți va fi pierdut). Registrele de recepție și de transmisie ale portului serial sunt accesate ambele din registrul de funcții speciale SBUF. Scrierea în SBUF încarcă registrul transmițător, iar citirea SBUF accesează un registru receptor, **fizic separat(!)**

Portul serial poate funcționa în 4 moduri:

Modul 0. Data serială intră și iese prin Rx/D. Tx/D furnizează clockul de deplasare. 8 biți sunt transmiși / recepționați (încapând cu LSB - cel mai puțin semnificativ bit). Rata de transfer este fixată la 1/12 din frecvența oscilatorului.

Modul 1. Se transmit 10 biți (prin Tx/D) sau se recepționează (prin Rx/D): un bit de start (0), 8 biți de date (începând cu cel mai puțin semnificativ - LSB), și un bit de stop (1). La recepție bitul de stop merge în RB8 din registrul de funcții speciale SCON. Rata transferului este variabilă.

Modul 2. Se transmit 11 biți (prin Tx/D) sau se recepționează (prin Rx/D): bitul de start (0), 8 biți de date (începând cu cel mai puțin semnificativ - LSB), un al 9-lea bit de date programabil și un bit de stop (1). În modul transmisie, cel de-al 9-lea bit de date (TB8 în SCON) poate să aibă atribuită valoarea 0 sau 1. Sau, de exemplu, bitul de paritate (P din PSW) poate fi transferat în TB8. În modul recepție, al 9-lea bit de date merge în RB8 din registrul de funcții speciale SCON, în timp ce bitul de stop este ignorat. Rata de transfer este programabilă la oricare din valorile 1/32 sau 1/64 din frecvența oscilatorului.

Modul 3. Se transmit 11 biți (prin Tx/D) sau se recepționează (prin Rx/D): un bit de start (0), 8 biți de date (începând cu bitul cel mai puțin semnificativ - LSB), un al 9-lea bit de date programabil și un bit de stop (1). De fapt, modul 3 este același ca și modul 2 în toate privințele, mai puțin rata de transfer. În modul 3, rata de transfer este variabilă.

Comunicații multiprocesor

Modurile 2 și 3 au o rezervă specială pentru comunicații multiprocesor. În aceste moduri se recepționează 9 biți de date. Cel de-al 9-lea bit merge în RB8. După aceasta, sosește un bit de stop. Portul poate fi programat astfel încât, atunci când se recepționează bitul stop, întreruperea portului serial va fi activată numai dacă RB8 = 1. Această proprietate se activează prin setarea bitului SM2 din SCON. În cele ce urmează se prezintă o modalitate de utilizare a acestei caracteristici în sisteme multiprocesor:

Când procesorul **master** dorește să transmită un bloc de date către unul sau mai mulți **slaves**, mai întâi transmite la ieșire un cuvânt de adresă care identifică slave-ul țintă. Un cuvânt de adresă diferă de cuvântul de date prin aceea că cel de-al 9-lea bit este "1" pentru cuvântul adresă și "0" pentru cuvântul de date. *Cu SM2 = 1, nici un slave nu va fi întrerupt de un cuvânt de date.* Un cuvânt de adresă, totuși, va întrerupe toți slave-ii, astfel încât fiecare slave poate examina cuvântul recepționat și să vadă dacă nu îi este adresat. Slave-ii adresați vor șterge biții lor SM2 și pregătesc recepția cuvintelor de date care vor veni. Slave-ii care nu sunt adresați, lasă biții lor SM2 setați și revin la problemele lor, ignorând cuvintele de date care sosesc.

SM2 nu are nici un efect în modul 0, iar în modul 1 poate fi folosit pentru a verifica valabilitatea bitului stop. În modul 1 recepție, dacă SM2 = 1, întreruperea de recepție nu va activa decât dacă se primește un bit de stop.

Registrul de control al portului serial

Registrul de control și de stare a portului serial este registrul funcției speciale SCON, prezentat mai jos. Acest registru conține nu numai biții de selecție a modului, dar și cel de-al 9-lea bit de date pentru transmitere și recepție (TB8 și RB8) și biții de întrerupere ai portului serial (TI și RI).

MSB				LSB			
SM0	SM1	SM2	REN	TB8	RB8	TI	RI

SM0	SM1	Mod	Descriere	Viteză (rata Baud)
0	0	0	registru de deplasare	$f_{osc} / 12$
0	1	1	UART de 8 biți	variabilă
1	0	2	UART de 9 biți	$f_{osc} / 64$ sau $f_{osc} / 32$
1	1	3	UART de 9 biți	variabilă

SM2	Activează comunicațiile multiprocesor în modurile 2 și 3. În modurile 2 sau 3, dacă SM2 este setat la 1, atunci RI nu va fi activat dacă cel de-al 9-lea bit de date (RB8) este 0. În mod 1, dacă SM2 = 1 atunci RI nu va fi activat dacă nu se recepționează un bit de stop valabil. În mod 0 SM2 trebuie să fie 0.
REN	Activează recepția serială. Setat prin software pentru a activa recepția. Șters prin software pentru a dezactiva recepția.
TB8	Al 9-lea bit de date ce va fi transmis în modurile 2 și 3. Setat sau șters prin software, după cum se dorește.
RB8	În modurile 2 și 3, este cel de-al 9-lea bit de date ce a fost recepționat. În mod 1, dacă SM2 = 0, RB8 este bitul de stop recepționat. În mod 0, RB8 nu este folosit
TI	Flagul întrerupere transmisie. Setat hardware la sfârșitul celui de-al 8-lea timp de bit în mod 0, sau la începutul bitului stop în celelalte moduri, în orice transmisie serială. Trebuie șters software.
RI	Flagul întrerupere recepție. Setat hardware la sfârșitul celui de-al 8-lea timp de bit în mod 0, sau la mijlocul timpului de bit stop în celelalte moduri, în orice transmisie serială (pentru excepție vezi SM2). Trebuie șters software.

Registrul de control al portului serial (SCON)

Rata de transfer

Rata de transfer (baud rate) în mod 0 este fixă: rata de transfer mod 0 = frecvența oscilatorului / 12. Rata de transfer în modul 2 depinde de valoarea bitului SMOD din registrul funcției speciale PCON. Dacă SMOD = 0 (care este valoarea la reset), rata de transfer este 1/64 din frecvența oscilatorului. Dacă SMOD = 1, rata de transfer este 1/32 din frecvența oscilatorului.

$$\text{Rata de transfer în mod 2} = \frac{2^{\text{SMOD}}}{64} \times \text{frecvența oscilatorului}$$

În circuitul 80C51, rata de transfer în modurile 1 și 3 este determinată de rata de depășire (divizare) a timerului 1.

Folosirea timerului 1 pentru generarea ratei de transfer

Când timerul 1 este folosit ca generator al ratei de transfer, atunci rata de transfer în modurile 1 și 3 este determinată de rata de depășire și de valoarea SMOD, după cum urmează:

$$\text{Rata de transfer în modurile 1 și 3} = \frac{2^{\text{SMOD}}}{32} \times (\text{rata de depasire a timerului 1})$$

Înteruperea timerului 1 trebuie **dezactivată** în această aplicație. Timerul/Numărător însuși poate fi configurat fie "timer", fie "numărător" și în fiecare din cele 4 moduri. În cele mai tipice aplicații, el este configurat pentru funcționare "timer", în modul cu *autoreîncărcare* (jumătatea *superioară* din TMOD = 0010'B'). În acest caz, rata transferului este dată de formula:

$$\text{Rata de transfer modurile 1,3} = \frac{2^{\text{SMOD}}}{32} \times \frac{\text{frecventa oscilatorului}}{12 \times [256 - (\text{TH1})]}$$

Se pot obține rate de transfer foarte scăzute cu timerul 1 lăsând activate înteruperile și configurând timerul pentru a funcționa ca un timer de 16 biți (jumătatea superioară din TMOD = 0001'B') și folosind înteruperile timerului 1 pentru a efectua o reîncărcare de 16 biți din software. În figura de mai jos sunt date diferite rate de transfer utilizate în mod obișnuit și felul în care se pot obține pornind de la timerul 1.

Rata Baud	f _{osc}	SMOD	Timer 1		
			C / \bar{T}	Mod	Valoare reîncărcare
Mod 0 max: 1.67 M	20 MHz	X	X	X	X
Mod 2 max: 625 k	20 MHz	1	X	X	X
Mod 1, 3 max: 104.2 k	20 MHz	1	0	2	FFH
19.2 k	11.059 MHz	1	0	2	FDH
9.6 k	11.059 MHz	0	0	2	FDH
4.8 k	11.059 MHz	0	0	2	FAH
2.4 k	11.059 MHz	0	0	2	F4H
1.2 k	11.059 MHz	0	0	2	E8H
137.5	11.059 MHz	0	0	2	1DH
110	6 MHz	0	0	2	72H
110	12 MHz	0	0	1	FEEBH

Rate Baud folosite în mod curent generate de timerul 1

Aplicație:

În cele ce urmează este prezentat un program de transmisie pe serială a ceea ce se găsește în registrul Acumulator. Programul utilizează Timerul T1 în mod cu auto-reîncărcare pentru generarea ratelor Boud. Se observă că programul așteaptă într-o buclă (unde aprinde și stinge un LED) până se termină de transmis cuvântul din SBUF (mai corect este, până când Timerul T1 programat anterior pe una din „ratele Boud”, cere o „înterupere” – setarea flagului TI).

```
name    transmisie_uart_0
org 0000H
jmp start

org 0023H
JMP sendchar

org 0040H
start:
    led    equ p3.0
    mov    TMOD,#00100000B ;programare Timer T1: 8 bit cu auto-incarcare din TH1
    mov    TL1,#1DH        ;INCARC OCTETUL INFERIOR AL TIMERULUI T1 137,5 BOUD RATE !!!!!!!
    mov    TH1,#1DH        ; 137,5 BOUD RATE !!!!!!! (vezi tabelul BOUD RATE)
    mov    SCON,#01000000b ;SETEZ UART variabil in MOD 1
    setb   EA              ;activeaza intreruperile globale
    setb   ES              ;activeaza intreruperea pe seriala
    setb   TR1            ;PORNEȘC timerul T1
    mov    A, #0          ;muta in Acumulator constanta zero
    setb   TI             ;FLAG transmisie intrerupere seriala

main:
    setb   led            ;bucla principala - aprinde si stinge in LED
    call   delay
    clr    led
    call   delay
    jmp    main           ;asteapta in bucla pana cand TI devine setat iar asta
                                ;se intampla doar cand e terminat de transmis tot ce era in SBUF

sendchar:
    clr    TI             ;sterge soft TI, flag setat de la terminarea transmisiei anterioare
    mov    SBUF, a        ;incepand din acest moment are loc transmisia pe seriala a lui SBUF
    inc    a              ;incrementez Acumulatorul pentru a transmite un nou caracter ASCII
    reti                 ;revenire din bucla

delay:
    mov    R0,#4          ;bucla de intarziere de 0,5sec.
    mov    R1,#208
    mov    R2,#184

delay0:
    djnz  R2,delay0
    djnz  R1,delay0
    djnz  R0,delay0
    ret

end                       ;terminare program
```